

# Implementation of Mason's Model on Circuit Analysis Programs

STEVEN A. MORRIS, MEMBER, IEEE, AND CHRISWELL G. HUTCHENS, MEMBER, IEEE

**Abstract**—The Redwood version of Mason's model has been rearranged to provide a topology suitable for computer modeling. The modified circuit uses a hybrid representation of the electromechanical transformer, an accurate approximation of the negative capacitor  $C_0$ , and modified coupling from the transformer to the acoustic transmission line. This arrangement provides a model that is applicable to circuit simulation packages. Model parameters are easily estimated directly from published data. This model, when implemented on a circuit-simulation program, is shown to be a useful design aid to determine acoustic transducer performance characteristics. The model allows the straightforward simulation of electrical matching, acoustic matching, transfer characteristics, bandwidth, impedance plots, and transient responses. The transmit-and-receive electronics and transducer with matching and backing are easily simulated.

## I. INTRODUCTION

**A**N IMPORTANT problem when designing ultrasonic transducers is the simulation of possible transducer configurations prior to construction. The designer would like to simulate the possible effects of backing materials, quarter-wave layers, electrical matching schemes and other design variables on the time response of the transducer prior to fabrication.

This paper shows how Mason's model can be implemented on a commercial circuit analysis program. This provides a powerful simulation tool for the transducer designer which can simulate quarter wave matching, electrical matching, and the associated driver-and-receiver electronics.

Equivalent circuits such as KLM [1] or Mason's original model [2] cannot be implemented on circuit analysis programs because of frequency dependent elements. Artificial transmission line circuits [3], [4] are overly complex and inaccurate at high frequency.

## II. THEORY

Fig. 1 shows a diagram explaining geometry and port definitions for a thickness-mode transducer along with Redwood's version of Mason's equivalent circuit [5]. The model consists of a capacitance  $C_0$ , a negative capacitance  $-C_0$ , an ideal transformer and a transmission line. The secondary of the transformer is attached to the shield of the transmission line, the shield having no inductance




Fig. 1. (a) Diagram of thickness-mode transducer. (b) Equivalent circuit of Mason as adapted by Redwood.

along its length.  $F_1$ ,  $F_2$  and  $v_1$ ,  $v_2$  are the forces and particle velocities at the crystal faces.

The SPICE transmission line model cannot be used to directly model the circuit of Fig. 1(b) because of the need for a shield with no inductance. By redrawing the circuit and redefining  $F_1$ ,  $F_2$ ,  $v_1$ , and  $v_2$  (Fig. 2), the SPICE transmission line model which contains inductance in both conductors can be used.

Fig. 3 shows how two other elements of the equivalent circuit are modeled. Fig. 3(a) is the hybrid representation of an ideal transformer using linear dependent sources, while Fig. 3(b) shows an approximation of the negative capacitance  $-C_0$ . Solving the nodal equations of Fig. 3(b) gives

$$I_C = -C_0 \frac{dV_C}{dt} \left[ 1 - \frac{C_s}{C_0} \right]$$

or

Manuscript received April 1, 1985; revised July 5, 1985.  
 S. A. Morris is with Amoco Production Co., 4502 East 41 Street, Tulsa, OK 74102, USA.  
 C. G. Hutchens is with the Department of Electrical Engineering, University of Tulsa, Tulsa, OK 74104, USA.  
 IEEE Log Number 8407233.




Fig. 2. Modification of model of Fig. 1(b) to incorporate a true transmission line.




Fig. 3. (a) Model of ideal transformer. (b) Approximation of negative capacitance.

$$I_C = -C_0 \frac{dV_C}{dt}, \quad \text{when } C_0 \gg C_s.$$

Fig. 4(a) shows the circuit of Fig. 2 as implemented for SPICE. VS1 and VS2 are zero valued sources used by dependent sources FCO and FCFMR as required by SPICE and have no effect on the circuit. R1, R2, R3, and R4 are needed to fulfill the SPICE requirement that every node have a dc path to ground. These resistors are chosen large enough that they have negligible effect on the circuit. R1 can be selected to reflect true dielectric losses if so desired. The SPICE deck is shown in Fig. 4(b) with the modifiable parameters (in brackets) and appropriate model equations, where  $Z_0$  is the crystal impedance and  $\tau$  is computed from the stiffened velocity.

The implementation of the model given here can be used with any circuit analysis program containing a transmission line model and suitable linear dependent sources.

### III. EXAMPLE

An example of the value of the SPICE implementation of the transducer model may be demonstrated by duplicating the simulation of Hayward and Jackson [6]. This particular simulation makes an excellent benchmark for accuracy because of the high frequencies present in the output waveform and the excellent experimental verification.

The properties of the transducer used by Hayward and Jackson are as follows:

Material	PZT-5A
Front load	water (1.5E6 rayl)
Back load	epoxy resin (9.1E6 rayl)

Acoustic transit time	430 ns
Diameter	20 mm.

From Berlincourt *et al.* [7], the values for the material properties of PZT-5A are


$$\begin{aligned} \rho &= 7.75E3 \text{ kg/m}^3 \\ \epsilon^s/\epsilon_0 &= 830 \\ e_{33} &= 15.8 \text{ coul/m}^2 \\ \nu^D &= 4350 \text{ m/s.} \end{aligned}$$

The parameters calculated for the equivalent circuit are

$$\begin{aligned} Z_0 &= \rho \nu^D A = 10.58 \text{ N/(m/s)} \\ \tau &= 430 \text{ ns} \\ C_0 &= \epsilon^s A / \tau \nu^D = 1230 \text{ pF} \\ hC_0 &= e_{33} C_0 / \epsilon^s = 2.643 \text{ N/V} \\ R_f &= A(1.5E6) = 471 \text{ N/(m/s)} \\ R_b &= A(9.1E6) = 2860 \text{ N/(m/s)} \end{aligned}$$

Fig. 5 shows the transducer and associated circuit along with the corresponding SPICE input deck and voltage input, where  $R_f$  and  $R_b$  are the front and back acoustic impedances, respectively.

Fig. 6 shows the transmitter response as generated by SPICE along with the simulation of Hayward and Jackson. The minor disagreement in the amplitude of the two waveforms is due to the fact that the SPICE parameters were derived from published data. Parameters taken from



(a)

```
.SUBCIRCUIT XDCR 3 2 1 7
CO 3 0 {CO}
R1 3 0 1MEG
VS1 4 3
CS 5 4 {.01*CO}
FCO 3 5 VS1 100
*
R2 5 0 1MEG
FXFMR 0 5 VS2 {hCO}
EXFMR 6 0 5 0 {hCO}
VS2 7 6
TMECH 1 0 2 0 {Z0}{tau}
R3 2 0 1 MEG
R4 1 0 1 MEG
.ENDS
```

Parameter Equations

$$C_0 = \frac{\epsilon^* A}{d}$$

$$Z_0 = \rho v^D A$$


$$\tau = \frac{d}{v^D}$$

$h$  may be calculated from

$$h = \frac{\epsilon^{.33}}{\epsilon^*}$$

(b)

Fig. 4. (a) SPICE implementation of model of Fig. 2. (b) SPICE subcircuit.



```
*
VIN 1 0 PULSE(300 0 0 100NSEC)
C1 1 2 2NF
R1 2 0 100
RBACK 4 5 2.8AK
RFRONT 3 5 471
X1 2 3 4 5 XDCR
.TRAN 50NSEC 4.5USEC
.PLOT TRAN V(3,5) (-20,20)
*
.SUBCKT XDCR 3 2 1 7
CO 3 0 1230PF
R1 3 0 1MEG
*
VS1 4 3
CS 5 4 12.3PF
FCO 3 5 VS1 100
*
R2 5 0 1MEG
FXFMR 0 5 VS2 2.643
EXFMR 6 0 5 0 2.643
VS2 7 6
*
TMECH 1 0 2 0 Z0=10.58K TD=430NSEC
R3 2 0 1MEG
R4 1 0 1MEG
.ENDS
*
.END
```

Fig. 5. Simulation of Hayward and Jackson's transducer.




Fig. 6. Simulated time response of circuit of Fig. 5

Hayward's experimental measurements were unavailable for SPICE simulation.

### V. CONCLUSION

The SPICE model given here is very useful for several reasons. Given a particular transducer, SPICE can be used to simulate the important characteristics of the transducer, e.g. electrical and acoustic input impedance, port-to-port transfer functions, receiver time response, and drive time response. Matching schemes are easily simulated using the SPICE program. Electrical matching is done by simply entering the matching network in the circuit description. Acoustic matching is simulated by adding a transmission line of suitable characteristic impedance and time delay between the front face and the load. Pulse-echo simulation is done by cascading two transducer circuits using a dependent voltage source. As demonstrated here, the accuracy is well beyond that required for most applications.

### REFERENCES

- [1] Krimholtz, Leedom, and Matthaei, "New equivalent circuit for elementary piezoelectric transducers," *Electron. Lett.*, vol. 6, no. 13, p. 338-339, June 27, 1970.

- [2] W. P. Mason, *Electromechanical Transducers and Wave Filters*. New York: D. Van Nostrand, 1942.
- [3] Schwarz, "Digital computer simulation of a piezoelectric thickness vibrator," *J. Acous. Soc. Amer.*, vol. 62, no. 2, pp. 463-467, Aug. 1977.
- [4] C. G. Hutchens and S. A. Morris, "A three-port model for thickness mode transducers using SPICE II," in *Proc. 1984 IEEE Ultrason. Symp.*, 1984, pp. 897-902.
- [5] M. Redwood, "Transient performance of a piezoelectric transducer," *J. Acoust. Soc. Amer.*, vol. 33, no. 4, pp. 527-536, Apr. 1961.
- [6] G. Hayward and M. Jackson, "Discrete-time modeling of the thickness mode piezoelectric transducer," *IEEE Trans. Sonics Ultrason.*, vol. SU-31, no. 3, May 1984.
- [7] D. A. Berlincourt, D. R. Corran, and H. Jaffe, "Piezoelectric and piezomagnetic materials and their function in transducers," *Physical Acoustics*, W. P. Mason, Ed. New York: Academic, 1964.



Steven Morris (S'83-M'84) was born on July 15, 1955, in Tulsa, OK. He received the B.S. and M.S. degrees in electrical engineering in 1982 and 1985, respectively, from Tulsa University.

Since 1982 he has been with the Amoco Production Company Research Center in Tulsa, where he is working on ultrasonic well-logging instruments.



Chriswell G. Hutchens (S'71-M'73) received the B.S. and M.S. degrees in electrical engineering from South Dakota State University, Brookings. He received the Ph.D. degree from the University of Missouri, Columbia, in 1979.

In 1980 he joined the faculty of the Department of Electrical Engineering at the University of Tulsa, OK. His current research interests include: circuit simulation models, microcomputer/analog instrumentation, digital integrated circuit design, and acoustic transducer design.

Dr. Hutchens is an experienced analog and microprocessor circuit designer. He has several years experience in medical electronics and is a Certified Clinical Engineer. He has served as a member of the Board of Clinical Engineering Certification and is a member and student faculty advisor of *Eta Kappa Nu*.